
A puzzle: C’s printf in Haskell

Here is a session that illustrates the puzzle—we renamed printf to format .

Main〉 : type format (Lit "Richard")
String
Main〉 format (Lit "Richard")
"Richard"

Main〉 : type format Int
Int → String
Main〉 format Int 60
"60"

Main〉 : type format (String :^: Lit " is " :^: Int)
String → Int → String
Main〉 format (String :^: Lit " is " :^: Int) "Richard" 60
"Richard is 60"

NB. ‘Main〉 ’ is the prompt; ‘: type’ displays the type of an expression.

Introducing phantom types

Suppose you want to embed a simple expression language in Haskell.

You are a firm believer of static typing, so you would like your embedded language
to be statically typed, as well.

☞ This rules out a simple Term data type as this choice would allow us to
freely mix terms of different types.

Idea: parameterize the Term type so that Term τ comprises only terms of type τ .

Zero :: Term Int
Succ,Pred :: Term Int → Term Int
IsZero :: Term Int → Term Bool
If :: ∀α .Term Bool → Term α→ Term α→ Term α

Alas, the signature above cannot be translated into a data declaration.

Introducing phantom types—continued

Idea: augment the data construct by type equations that allow us to constrain
the type argument of Term.

data Term τ = Zero with τ = Int
| Succ (Term Int) with τ = Int
| Pred (Term Int) with τ = Int
| IsZero (Term Int) with τ = Bool
| If (Term Bool) (Term α) (Term α) with τ = α

Thus, Zero has Type τ with the additional constraint τ = Int .

NB. The type variable α does not appear on the left-hand side; it can be seen as
being existentially quantified.

Introducing phantom types—continued

Here is a simple interpreter for the expression language. Its definition proceeds by
straightforward structural recursion.

eval :: ∀τ .Term τ → τ
eval (Zero) = 0
eval (Succ e) = eval e + 1
eval (Pred e) = eval e − 1
eval (IsZero e) = eval e 0
eval (If e1 e2 e3) = if eval e1 then eval e2 else eval e3

Even though eval is assigned the type ∀τ .Term τ → τ , each equation has a
more specific type as dictated by the type constraints.

☞ The interpreter is quite noticeable in that it is tag free. If it receives a
Boolean expression, then it returns a Boolean.



6 JJ J I II 2

Introducing phantom types—continued

Here is a short interactive session that shows the interpreter in action.

Main〉 let one = Succ Zero
Main〉 : type one
Term Int
Main〉 eval one
1

Main〉 eval (IsZero one)
False

Main〉 IsZero (IsZero one)
Type error: couldn’t match ‘Int’ against ‘Bool’

Introducing phantom types—continued

The type Term τ is quite unusual.

I Term is not a container type: an element of Term Int is an expression that
evaluates to an integer; it is not a data structure that contains integers.

I We cannot define a mapping function (α→ β)→ (Term α→ Term β) as
for many other data types.

I The type Term β might not even be inhabited: there are, for instance, no
terms of type Term String .

Since the type argument of Term is not related to any component, we call Term
a phantom type.

Generic functions

We can use phantom types to implement generic functions, functions that work
for a family of types.

Basic idea: define a data type whose elements represent types.

data Type τ = RInt with τ = Int
| RChar with τ = Char
| RList (Type α) with τ = [α ]
| RPair (Type α) (Type β) with τ = (α, β)

rString :: Type String
rString = RList RChar

☞ An element rt of type Type τ is a representation of τ .

Generic functions—continued

A useful generic function is compress which compresses data to string of bits.

Main〉 : type compress RInt
Int → [Bit ]
Main〉 compress RInt 60
<00111100000000000000000000000000>

Main〉 : type compress rString
[Char ]→ [Bit ]
Main〉 compress rString "Richard"
<101001011100101111100011100010111100001110100111100100110>

Generic functions—continued

The generic function compress pattern matches on the type representation and
then takes the appropriate action.

data Bit = 0 | 1

compress :: ∀τ .Type τ → τ → [Bit ]
compress (RInt) i = compressInt i
compress (RChar ) c = compressChar c
compress (RList ra) [ ] = 0 : [ ]
compress (RList ra) (a : as) = 1 : compress ra a

++ compress (RList ra) as
compress (RPair ra rb) (a, b) = compress ra a

++compress rb b

NB. We assume that compressInt :: Int → [Bit ] and
compressChar :: Char → [Bit ] are given.

Dynamic values

Using the type of type representations we can also implement dynamic values.

data Dynamic = Dyn (Type τ ) τ

☞ A dynamic value is a pair consisting of a type representation of Type τ and a
value of type τ for some type τ .

To be able to form dynamic values that contain dynamic values (for instance, a
list of dynamics), we add Dynamic to Type τ .

data Type τ = · · ·
| RDyn with τ = Dynamic

☞ Type and Dynamic are now defined by mutual recursion.

Dynamic values—continued

It is not difficult to extend compress so that it also works for dynamic values: a
dynamic value contains a type representation, which compress requires as a first
argument.

. . .
compress RDyn (Dyn ra a) = compressRep (Rep ra) ++ compress ra a

Exercise: Implement the function compressRep that compresses a type
representation.

data Rep = Rep (Type τ )

compressRep :: Rep → [Bit ]

Dynamic values—continued

The following session illustrates the use of dynamics and generics.

Main〉 let ds = [Dyn RInt 60,Dyn rString "Bird" ]
Main〉 : type ds
[Dynamic ]

Main〉 Dyn (RList RDyn) ds
Dyn (RList RDyn) [Dyn RInt 60,Dyn (RList RChar ) "Bird" ]

Main〉 compress RDyn it
<01010010000011110000000000000000000000000010100011010000
111001011101001111001001100>
Main〉 uncompress RDyn it
Dyn (RList RDyn) [Dyn RInt 60,Dyn (RList RChar ) "Bird" ]

NB. it always refers to the previously evaluated expression.

Dynamic values—continued

Turning a dynamic value into a static value involves a dynamic type check.

tequal :: ∀τ ν .Type τ → Type ν → Maybe (τ → ν)
tequal (RInt) (RInt) = return id
tequal (RChar ) (RChar ) = return id
tequal (RList ra1) (RList ra2)

= liftM list (tequal ra1 ra2)
tequal (RPair ra1 rb1) (RPair ra2 rb2)

= liftM2 pair (tequal ra1 ra2) (tequal rb1 rb2)
tequal = fail "types are not equal".

NB. The functions list and pair are the mapping functions of the list and the
pair type constructor.

☞ ‘tequal ’ can be made more general and more efficient!

Dynamic values—continued

The function cast transforms a dynamic value into a static value of a given type.

cast :: ∀τ .Dynamic → Type τ → Maybe τ
cast (Dyn ra a) rt = fmap (λf → f a) (tequal ra rt)

Here is a short session that illustrates its use.

Main〉 let d = Dyn RInt 60
Main〉 cast d RInt
Just 60
Main〉 cast d RChar
Nothing

Generic traversals

☞ Generic functions are first-class citizens.

Let us illustrate this point by implementing a small combinator library for so-called
generic traversals.

type Name = String
type Age = Int
data Person = Person Name Age

data Type τ = · · ·
| RPerson with τ = Person

Generic traversals—continued

The function tick s is an ad-hoc traversal—Traversal will be defined shortly.

tick :: Name → Traversal
tick s (RPerson) (Person n a)
| s n = Person n (a + 1)

tick s rt t = t

The following session shows tick in action.

Main〉 let ps = [Person "Norma" 50,Person "Richard" 59]

Main〉 everywhere (tick "Richard") (RList RPerson) ps
[Person "Norma" 50,Person "Richard" 60]

☞ everywhere applies its argument ‘everywhere’ in a given value.

Generic traversals—continued

A generic traversal takes a type representation and transforms a value of the
specified type.

type Traversal = ∀τ .Type τ → τ → τ.

☞ The universal quantifier makes explicit that the function works for all
representable types.

Here is a tiny ‘traversal algebra’.

copy :: Traversal
copy rt = id

(◦) :: Traversal → Traversal → Traversal
(f ◦ g) rt = f rt · g rt

Generic traversals—continued

The everywhere combinator is implemented in two steps.

First, we define a function that applies a traversal f to the immediate components
of a value: C t1 . . . tn is mapped to C (f rt1 t1) . . . (f rtn tn) where rt i is the
representation of ti’s type.

imap :: Traversal → Traversal
imap f (RInt) i = i
imap f (RChar ) c = c
imap f (RList ra) [ ] = [ ]
imap f (RList ra) (a : as) = f ra a : f (RList ra) as
imap f (RPair ra rb) (a, b) = (f ra a, f rb b)
imap f (RPerson) (Person n a) = Person (f rString n) (f RInt a)

☞ imap can be seen as a ‘traversal transformer’.

Generic traversals—continued

Second, we tie the recursive knot.

everywhere, everywhere ′ :: Traversal → Traversal
everywhere f = f ◦ imap (everywhere f )
everywhere ′ f = imap (everywhere ′ f ) ◦ f

☞ everywhere f applies f after the recursive calls (it proceeds bottom-up),
whereas everywhere ′ applies f before (it proceeds top-down).

Functional unparsing

Recall the printf puzzle.

Main〉 : type format (Lit "Richard")
String
Main〉 format (Lit "Richard")
"Richard"

Main〉 : type format Int
Int → String
Main〉 format Int 60
"60"

Main〉 : type format (String :^: Lit " is " :^: Int)
String → Int → String
Main〉 format (String :^: Lit " is " :^: Int) "Richard" 60
"Richard is 60"

Functional unparsing—first try

Obvious idea: turn the type of directives, Dir , into a phantom type so that

format :: ∀θ .Dir θ → θ

(String :^: Lit " is " :^: Int) :: Dir (String → Int → String).

The format directive can be seen as a binary tree of type representations: Lit s ,
Int , String form the leaves, ‘:^:’ constructs the inner nodes.

The type of format is obtained by linearizing the binary tree.

Functional unparsing—first try

The types of Lit , Int , and String are immediate.

Lit :: String → Dir String
Int :: Dir (Int → String)
String :: Dir (String → String)

The type of ‘:^:’ is more involved.

(:^:) :: ∀θ ρ .Dir θ → Dir ρ→ Dir (θ( ρ)

String ( ν = ν
(α→ τ )( ν = α→ (τ ( ν)

NB. Read ‘(’ as concatenation, String as the empty list, and ‘→’ as cons.

☞ Alas, on the type level we can only use cons, not concatenation.

Accumulating parameters

Fortunately, Richard knows how to get rid of concatenation, see IFPH 7.3.1.

data Btree α = Leaf α | Fork (Btree α) (Btree α)

flatten :: ∀α .Btree α→ [α ]
flatten t = flatcat t [ ]

flatcat :: ∀α .Btree α→ [α ]→ [α ]
flatcat (Leaf a) as = a : as
flatcat (Fork tl tr ) as = flatcat tl (flatcat tr as)

The helper function flatcat enjoys

flatten t = x ≡ ∀as .flatcat t as = x ++ as .

Functional unparsing—second try

☞ Add an accumulating parameter to Dir .

Lit :: String → ∀θ .DirCat θ θ
Int :: ∀θ .DirCat θ (Int → θ)
String :: ∀θ .DirCat θ (String → θ)

The data type DirCat enjoys

e :: Dir τ ≡ e :: ∀θ .DirCat θ (τ ( θ).

The constructor ‘:^:’ realizes type composition.

(:^:) :: ∀θ1 θ2 θ3 .DirCat θ2 θ3 → DirCat θ1 θ2 → DirCat θ1 θ3

Functional unparsing—second try—continued

Now, let’s tackle the definition of format .

format :: ∀θ .DirCat String θ → θ
format (Lit s) = s
format (Int) = λi → show i
format (String) = λs → s
format (d1 :^: d2) = ?

☞ The type of format is not general enough to push the recursion through.

Functional unparsing—third try

☞ Fortunately, continuations save the day.

format ′ :: ∀θ ρ .DirCat θ ρ→ (String → θ)→ (String → ρ)
format ′ (Lit s) = λcont out → cont (out ++ s)
format ′ (Int) = λcont out → λi → cont (out ++ show i)
format ′ (String) = λcont out → λs → cont (out ++ s)
format ′ (d1 :^: d2) = λcont out → format ′ d1 (format ′ d2 cont) out

The helper function takes a continuation and an accumulating string argument.

format :: ∀ρ .DirCat String ρ→ ρ
format d = format ′ d id ""

Functional unparsing—third try—continued

Ouch, format ′ has a quadratic running time.

But again, Richard knows how to cure this deficiency . . .


